

Building Blocks of Beautiful RTF Destination Reports: Using Microsoft Word's Heading System to Add an Outline Content Pane

Debra A. Goldman, MS, Memorial Sloan Kettering Cancer Center

ABSTRACT

Adding layers of organization to reports is a beneficial communication tool, particularly for lengthy reports or for client-programmer relationships where in person meetings are not frequent. One reason SAS users may prefer to use PDF destination over RTF output is the automatic left hand outline of PROCedures in PDFs. These bookmarks make it easy to see what's in the report, to see where one is within the report, and to seamlessly move in between sections.

Fortunately for those that want to or be required to have their reports to be in Microsoft Word, the heading feature allows us to create a similar content pane. This pane can be clicked through and tracks where one is in the document as one moves through. In a prior WUSS paper¹, I reviewed options for adding Table of Contents and Hyperlinks to one's RTF reports. It also introduced the concept of using RTF code within SAS® to make use of Microsoft Word features. This paper will show users how to add multiple layers of headings in a navigable left side panel to one's RTF reports. Users should be familiar with ODS RTF destination, and the content is for users of SAS 9.4 or higher and Microsoft Office 2010 or higher.

INTRODUCTION

I discovered the ability to add headers inadvertently while looking at a colleague's report from RMarkdown. I noticed that her section titles also appeared on the left hand side in a panel. Knowing that it could be done in R, meant that R was using RTF syntax; therefore, I knew the same operation could be performed in SAS.

While headings have existed in Microsoft Word for some time, the docked navigation pane, illustrated below, was added for MS Office 2010. In previous versions, the heading navigation system was known as "Document Map". Headings are both a style of text presentation and also a section index. It can be viewed by going to the View tab and checking "Navigation Pane" or by typing CTRL+F and clicking on "Headings".




Figure 1. Illustration of navigation pane

The same headings that appear in Figure 1, such as "Abstract" and "Introduction", can be added using SAS. This paper will demonstrate how to create a customized navigation pane within SAS for your reports.

I will be using the SASHELP dataset, SASHELP.HEART, throughout this paper. This paper is intended for those who use the ODS RTF destination.

EXAMPLE DATASET

SASHELP.HEART contains data on 5,209 patients from the Framingham Heart Study. Please see SASHELP documentation for more information.²

ODS TEXT = AND PROC ODSTEXT

ODS TEXT = and PROC ODSTEXT are how we add text to the ODS destination. Prior WUSS papers^{1,3} introduced paragraphs, lists, and single line text. Below, we'll briefly cover how to modify formatting in both procedures. Please see the references for further details on adding text to ODS RTF documents.

ODS TEXT

The basic syntax for ODS TEXT is simple:

```
ods text="Table 1. Patient and Clinical Characteristics";
```

Table 1. Patient and Clinical Characteristics

Output 1. Basic ODS text output

SAS inserts a 1x1 dimension table with invisible borders containing the text written. As ODS TEXT is not a procedure, no `run` statement is required. The formatting of the text can be done the document within `PROC TEMPLATE`, by changing the formatting with a call of the `ods escapechar` followed by a style call, or by using RTF syntax. We'll cover the RTF syntax in the next section.

Option 1. Change format of text within PROC TEMPLATE

```
proc template;  
define style styles.wuss2019;  
parent=styles.sasdocprinter;  
class usertext from systemtitle / font=("Arial",11pt);  
style paragraph from usertext;  
...  
end;  
run;
```

SAS Syntax	RTF/SAS Function
<code>class usertext from systemtitle / font=("Arial",11pt);</code> <code>style paragraph from usertext;</code>	Tells SAS appearance of text that appears in documents using this style

Table 1. Explanation of PROC TEMPLATE Code

Option 2. Changing format of text with style statement

```
ods escapechar="^";  
ods text="^S={fontweight=bold}Table 1. Patient and Clinical Characteristics";
```

SAS Syntax	RTF/SAS Function
<code>ods escapechar="^";</code>	Tells SAS which character will be used to escape from current procedure into another operation
<code>^S={fontweight=bold}</code>	Calls SAS style and modifies font weight to bold

Table 2. Explanation of format change code using an escape character

Table 1. Patient and Clinical Characteristics

Output 2. Example of ODS text line in bold

PROC ODSTEXT

PROC ODSTEXT allows us to enter paragraphs or lists of text. Below are two code chunk examples of the PROC ODSTEXT used to make the report for this paper. Formatting can be set within PROC TEMPLATE, within each specific paragraph, within each list, or within each item of the list using a style statement in options.

```
proc odstext;  
p "{\pard\s1 Table of Contents \par}" / style=[fontsize=11pt fontweight=bold  
fontfamily=Arial];  
list / style={liststyletype="decimal" fontsize=11pt};  
item "{Statistical Methods, Page} {\field {\*\fldinst PAGeref Methods \h}}  
^n ^n"/style=[color=blue url='#Methods'];  
item "{Patient and Clinical Characteristics, Page} {\field {\*\fldinst  
PAGeref Table1 \h}} ^n ^n"/style=[ color=blue url='#Table1'];  
item "{Boxplot of Age by Cholesterol Status, Page} {\field {\*\fldinst  
PAGeref Figure1 \h}} ^n ^n"/style=[color=blue url='#Figure1'];  
end;  
run;
```

SAS Syntax	RTF/SAS Function
/style=[]	Modifies format of text

Table 3. Code for modifying format of text

Table of Contents

1. Statistical Methods, Page 2
2. Patient and Clinical Characteristics, Page 3
3. Boxplot of Age by Cholesterol Status, Page 4

Output 3. Example of list

```
proc odstext;  
p "{\pard\s1480\slmult1 \fi720 We summarized patient data stratified by  
gender using medians and ranges for continuous variables  
and frequencies and percentages for categorical variables. The relationship  
between age and cholesterol status for each gender was visualized with  
boxplots. \par}" /style=[fontsize=11pt fontfamily=Arial ];  
run;
```

Statistical Methods

We summarized patient data stratified by gender using medians and ranges for continuous variables and frequencies and percentages for categorical variables. The relationship between age and cholesterol status for each gender was visualized with boxplots.

Output 4. Example of paragraph text

RTF SYNTAX IN SAS

Using RTF syntax in SAS allows us to take advantage of underlying features of Word documents, such as indexing⁴, table of contents³, line spacing², and other formatting. For example, in the text above, we made use of bookmark fields and added page references through the syntax, `{\field {*\fldinst PAGEREF`, and we double spaced our text through the syntax, `\s1480\s1mult1 \fi720`. These techniques are just a few of the ways to enhance our SAS reports through the underlying output language.

As mentioned in the prior section, another way to modify formatting is to use RTF syntax. The command, `\b`, bolds text, which is equivalent to `^S={fontweight=bold}` and `\u1` will underline, which is equivalent to `^S={textdecoration=underline}`. This is the syntax needed to make the line appear below:

```
ods text="{\pard \b\u1 Table 1. Patient and Clinical Characteristics \par}";
```

Table 1. Patient and Clinical Characteristics
--

Output 5. Example of using RTF syntax to format

While the formatting can be changed with an escape character, RTF syntax will **not compile** if an escape character is called. Therefore, the above syntax is helpful for changing formatting in the context of other RTF calls.

SAS SYNTAX NEEDED FOR RTF CALLS

You may have noticed a few additional syntax pieces in the text above and in the PROC ODSTEXT section, specifically `{}`, `\pard`, and `\par`. Brackets allow for SAS to recognize the RTF syntax listed within:

```
ods text="{\pard\s1\b\u1 Statistical Methods \par}";
```

Additionally, the majority of RTF commands will need to appear within an RTF paragraph to be recognized:

```
ods text="{\pard\s1\b\u1 Statistical Methods \par}";
```

SAS Syntax	RTF/SAS Function
{ }	Surround text and allow for RTF syntax to be used
\pard \par	Tells Microsoft Word that the text is a paragraph and allows for certain RTF commands to be used

Table 2. Explanation of RTF syntax Code

Once we add the brackets and paragraph calls, RTF commands will be recognized and converted by SAS.

CREATE NAVIGATION PANE

Now that we've covered the basics of text and RTF syntax, we only need to add a few specific commands to add our headings. The same syntax will apply regardless of whether we're using `ods text` or PROC ODSTEXT.

STEP 1. ADD WORDSTYLE STATEMENT TO ODS RTF CALL

Within the ODS call a `wordstyle` statement needs to be included. This tells Microsoft Word that `\s1` will refer to Heading 1, `\s2` to Heading 2, etc.⁶ `S1` is the highest level of heading; `S2` will be embedded under `S1`, and `S3` will be embedded under `S2`. Importantly, no other abbreviation can be used; `\r1` or `\t1` will not work, as `\s#` is the RTF call for headings.

```
ods rtf file="..."
style=wuss2019 headery=320 startpage=no
wordstyle="{\s1 Heading 1 \s2 Heading 2 \s3 Heading 3;}";
```

SAS Syntax	RTF/SAS Function
<code>wordstyle="{\s1 Heading 1 \s2 Heading 2 \s3 Heading 3;}";</code>	Line of RTF code needed to tell SAS that \s1 will refer to Heading style 1,

Table 4. Explanation of PROC TEMPLATE Code

In the above word style, we've added calls three levels, S1, S2, and S3, but we could go out further to S4, S5 etc. It depends on how many layers of headings we want.


STEP 2. ADD HEADING SYNTAX TO EXISTING ODS TEXT

Now, we add the \s1 syntax to our existing text.

```
PROC ODSTEXT;
p "{\pard\s1 Table of Contents \par}" / style=[fontsize=11pt fontweight=bold
fontfamily=Arial];
...;
run;
ods text="{\pard\s1\b\ul Statistical Methods \par}";
proc odstext;
...;
ods text="{\pard\s1\b Table 1. Patient and Clinical Characteristics \par}";
proc report data=work.table_patientchar nowindows headline missing spanrows ;
...;
ods text="{\pard\s1\b Figure 1. Box plot of Age at Start by Cholesterol
Status for Each Gender \par}";
proc sgplot data=sashelp.heart;
...;
ods rtf close;
```

SAS Syntax	RTF/SAS Function
<code>\s1</code>	RTF syntax that indicates text provided will be Heading Level 1


Table 5. Explanation of RTF Syntax for Headings



Output 6. Navigation pane with all titles as S1 level

Here is what our navigation looks like with the above syntax. All the titles are on the same level. We can also embed headings. For example, we can embed Table 1 and Figure 1 under a section called “Analyses” using the following text:

```
PROC ODSTEXT;  
p "{\pard\s1 Table of Contents \par}" / style=[fontsize=11pt fontweight=bold  
fontfamily=Arial];  
...;  
run;  
ods text="{\pard\s1\b\ul Statistical Methods \par}";  
proc odstext;  
...;  
ods text="{\pard\s1\i Analyses \par}";  
ods text="{\pard\s2\b Table 1. Patient and Clinical Characteristics \par}";  
proc report data=work.table_patientchar nowindows headline missing spanrows ;  
...;  
ods text="{\pard\s2\b Figure 1. Box plot of Age at Start by Cholesterol  
Status for Each Gender \par}";  
proc sgplot data=sashelp.heart;  
  
...;
```



Output 7. Navigation pane with subheadings

Notice how Table 1 and Figure 1 are now embedded under analyses. All we have to do now is click on the text and it will take to the relevant section.

CONCLUSION

Headings can be used to supplement a table of contents or to separately highlight specific findings that are important for the reader to review. We put so much time and energy into running our analyses; adding RTF elements, such as headings, make it easier for others to follow along and understand what we've done. By incorporating it into our SAS syntax, our report remains reproducible with no work needed outside the SAS system. Well, you may have to show a user or two how to open the Navigation Pane.

REFERENCES

1. Goldman, D. A. 2018. "Connecting the Dots: Using ODS RTF ANCHORS, ODS RTF BOOKMARKS and Table of Contents to Guide Readers Through Your SAS® RTF Reports". Proceedings of the Western Users of SAS Software Conference 2018. Sacramento, CA. SAS Institute Inc., Available at: https://lexjansen.com/cgi-bin/xsl_transform.php?x=wuss2018#wuss2018.046.dp
2. SAS Institute Inc. 2016. *Sas help Data Sets*. Cary, NC: SAS Institute Inc.
3. Goldman, D. A. 2017. "Creating Complete Automated Reports in SAS using the ODS RTF Destination: Protect your Results from Human Error". Proceedings of the Western Users of SAS Software Conference 2017. Long Beach, CA. SAS Institute Inc., Available at: https://www.lexjansen.com/wuss/2017/32_Final_Paper_PDF.pdf
4. Small, E. 2007. "An RTF Document Index: Easier Than you Think". Proceedings of the North Eastern SAS Users Group. Baltimore, MD. SAS Institute Inc., Available at: <https://www.lexjansen.com/nesug/nesug07/cc/cc37.pdf>.
5. Axelrod, E. and Shamlin, D. 2004. "Skinning the Cat This Way and That: Using ODS to Create Word Documents That Work for You". Proceedings of the SAS Users Group International. Montreal, Canada. SAS Institute Inc., Available at: <https://support.sas.com/resources/papers/proceedings/proceedings/sugi29/084-29.pdf>.

ACKNOWLEDGMENTS

Thank you to Stephanie Lobaugh, MS and Karissa Whiting, MS, for providing feedback for this paper.

RECOMMENDED READING

- *SAS® Output Delivery System: Procedures Guide, Third Edition*

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Debra A. Goldman
Memorial Sloan Kettering Cancer Center
485 Lexington Avenue 2nd Floor
New York, NY 10017
646-888-8331
goldmand@mskcc.org
<https://www.mskcc.org/profile/debra-goldman>
<https://www.linkedin.com/in/debra-a-g-301265b/>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.